
OLTP "Interactive"
(5250 Workstation sessions)

Background Processing
(Batch jobs)

= 3 metodi:
®» Comando CL RTVJOBA
= APIQUSRJOBI
® Funzione SOL GET_JOB_INFO

Comando RTVJOBA

® Da un programma CL il metodo pitu immediato e utilizzare il
comando RTVJOBA

® Tra i vari attributi del job il parametro TYPE restituisce la tipologia

® "1’ perinterattivi

®» 'Q' perbatch

dcl &inter *char 1

rtvjoba type(&inter)

https://www.ibm.com/docs/api/v1/content/ssw_ibm_i_75/cl/rtvjoba.htm

[ da V1R3 ]

APIQUSRIJOBI/1

® |n un programma RPG e possibile richiamare 'API QUSRJOBI

® Come molte API la lista dei parametri puo essere molto lunga, ma

per il nostro scopo possiamo limitarci a richiedere il formato piu
ridotto JOBIO100

®» || sottocampo JobType (pos. 61) della struttura dati del formato
JOBIO100 vale'I' perijob interattivi

dcl-pr RtvJobA extpgm('QUSRJIOBI');
JobAttr char(32766) options(xvarsize); // receiver variable

RcvVarLen int(10) const; // byte retrieved in receiver variable
FormatName char(8) const; // format name
JobName char(26) const; // qualified job name
InternJobID char(16) const; // internal job identifier
end-pzr;

Power coffee - MKa -*- APIJOB_H.RPGLE, TRTVJOBA2.RPGLE



https://www.ibm.com/docs/en/i/7.5?topic=ssw_ibm_i_75/apis/qusrjobi.html

APl QUSRIJOBI/2

dcl-ds JOBIO100 qualified template;
ByteRet int(10); // bytes returned
BytesAvl int(10); // bytes available
JobName char(10); // Job Name
JobUser char(10); // User
JobNumber chazr(6); // Job number
JobInternID char(16); // Internal job ID
JobStatus char(10); // Job status

==> JobType char(1); // Job type ('I' interactive, 'B' batch)
JobSubtype char(l); // Job subtype
JobReservl char(2);
JobRunPty int(10); // Priorita esecuzione
JobTimeSlc int(10); // Time slice

\ JobDftWait int(10); // Default wait
% JobPurge char(10); // Purge ('%«YES'...)

end-ds;
dcl-c JOBA_BASICS 'JOBIO10O'; // Basic performance

RtvJobA(JobAttr : %size(JobAttr) : JOBA_BASICS

Funzione SQL GET_JOB_INFO

® | 3 funzione di tabella GET JOB INFO restituisce varie
informazioni su un job, tra cui nel campo V_ACTIVE_JOB_TYPE
il valore 'INT" perijob interattivi

dcl-s JobType varchar(3);

exec sqgl
select V_ACTIVE_JOB_TYPE
into :JobType
from table(GET_JOB_INFO('*"));

https://www.ibm.com/docs/en/i/7.5?topic=services-get-job-info-table-function

Confronto prestazioni

Reperimento del tipo job 10.000 volte consecutivamente

Durata(ms)

API ~ 43.000
comandoCL ~1.200.000 ~ 30 volte piu lento
funzione SQL ~120.000.000 ~ 2.800 volte piu lento

Welcome to PUB400.COM - Your public IBMi 7.5 server
by Holger Scherer (rzkh.de)

IBM 1 7.5
Power 9009-41G

Dispositivo processore: EP52
Ram: 50 Gb

What is PUB400.COM?

« Afree and public Server running IBM i 7.5 for everyone

= You can create your own user profile, you have 400MB of disk storage and two
private libraries

« You can program in CL, RPG, SQL, COBOL, train your skills and start learning
about the best server operating system ever ;)

« Create your own programming projects, you can even use node s or other web
technologies (contact us for http settings)

« An average of 25 new users per dayl Great IBM | community!

      /IF NOT DEFINED(APIJOB)

      /DEFINE APIJOB



       //--- PROTOTIPI API JOB MANAGEMENT --- START ---

       // (c) 2022 MarkOneTools - www.markonetools.it



       //**************************************************************************

       // Retrieve Job Attributes

       //

       // Only required parameter

       // Only format JOBI0100

       // https://www.ibm.com/docs/en/i/7.5?topic=ssw_ibm_i_75/apis/qusrjobi.html



       dcl-pr RtvJobA extpgm('QUSRJOBI');

         JobAttr char(32766) options(*varsize); // receiver variable

         RcvVarLen int(10) const;                // byte retrieved in receiver variable

         FormatName char(8) const;               // format name

         JobName char(26) const;                 // qualified job name

         InternJobID char(16) const;             // internal job identifier

       end-pr;



       // DS per mappare info restituite dal formato JOBI0100

       //  Basic performance information

       dcl-ds JOBI0100 qualified template;

         ByteRet int(10);        // bytes returned

         BytesAvl int(10);       // bytes available

         JobName char(10);       // Job Name

         JobUser char(10);       // User

         JobNumber char(6);      // Job number

         JobInternID char(16);   // Internal job ID

         JobStatus char(10);     // Job status

==>      JobType char(1);        // Job type ('I' interactive, 'B' batch)

         JobSubtype char(1);     // Job subtype

         JobReserv1 char(2);

         JobRunPty int(10);      // Priorità esecuzione

         JobTimeSlc int(10);     // Time slice

         JobDftWait int(10);     // Default wait

         JobPurge char(10);      // Purge ('*YES'...)

       end-ds;



       // API QUSRJOBI format

       dcl-c JOBA_BASICS    'JOBI0100'; // Basic performance

       dcl-c JOBA_ADDPFR    'JOBI0150'; // additional performance

       dcl-c JOBA_WRKACTJOB 'JOBI0200'; // wrkactjob info

       dcl-c JOBA_JOBQ      'JOBI0300'; // job queue and output queue

       dcl-c JOBA_JOBA      'JOBI0400'; // job attribute

       dcl-c JOBA_MSGLOG    'JOBI0500'; // message logging

       dcl-c JOBA_ACTJOB    'JOBI0600'; // active job

       dcl-c JOBA_LIBLIST   'JOBI0700'; // library list

       dcl-c JOBA_LIBLISTE  'JOBI0750'; // library list (extended)

       dcl-c JOBA_ACTJOBSGN 'JOBI0800'; // active job signal

       dcl-c JOBA_ACTJOBSQL 'JOBI0900'; // active job SQL

       dcl-c JOBA_PFRSTAT   'JOBI1000'; // elapsed performance statistics

      /ENDIF



      /IF NOT DEFINED(QNAMEJOB)

      /DEFINE QNAMEJOB

       // nome qualificato lavoro

       dcl-ds QNameJob qualified template;

         Name char(10); // nome job

         User char(10); // utente

         Nbr char(6);   // numero job

       end-ds;

      /ENDIF


/* TEST REPERIMENTO TIPO JOB */



/* (c) 2022 MarkOneTools - www.markonetools.it */



      pgm



      dcl        &inter *char 1



      rtvjoba    type(&inter)



      if         (&inter *eq '1') then(do)

         sndpgmmsg  msg('Il job è interattivo') topgmq(*prv (*)) +

               tomsgq(*hstlog)

      enddo

      else       do

         sndpgmmsg  msg('Il job è batch') topgmq(*prv (*)) tomsgq(*hstlog)

      enddo



      endpgm


**free





// REPERIMENTO TIPO JOB CON API QUSRJOBI

// (c) 2022 MarkOneTools - www.markonetools.it



ctl-opt copyright('MarkOneTools')

        decedit('0,')

        indent(' ')

        option(*nodebugio: *srcstmt: *showcpy)

        expropts(*resdecpos)

        extbinint(*yes);



/copy qprtsrc,apijob_h



// Formato informazioni restituite

dcl-s JobAFormat char(8);

// struttura dati informazioni restituite da API

dcl-ds JobAttr likeds(JOBI0100);



dcl-s Descrizione varchar(12);



// reperimento attributi lavoro corrente

RtvJobA(JobAttr : %size(JobAttr) : JOBA_BASICS : '*' : ' ');



if JobAttr.JobType = 'I';

  Descrizione = 'interattivo';

else;

  Descrizione = 'batch';

endif;



dsply ('Job ' +

         %trim(JobAttr.JobName) + '/' +

         %trim(JobAttr.JobUser) + '/' +

         JobAttr.JobNumber +

         ' è ' + JobAttr.JobType + ' ' + Descrizione) '*EXT';



*inlr = *on;


**free





// REPERIMENTO TIPO JOB CON FUNZIONE SQL GET_JOB_INFO

// (c) 2022 MarkOneTools - www.markonetools.it



ctl-opt copyright('MarkOneTools')

        decedit('0,')

        indent(' ')

        option(*nodebugio: *srcstmt: *showcpy)

        expropts(*resdecpos)

        extbinint(*yes);



dcl-s JobType varchar(3);

dcl-s Descrizione varchar(12);



clear JobType;

exec sql

  select V_ACTIVE_JOB_TYPE

    into :JobType

    from table(GET_JOB_INFO('*')) ;



if JobType = 'INT';

  Descrizione = 'interattivo';

else;

  Descrizione = 'batch';

endif;



dsply ('Il job è ' + JobType + ' ' + Descrizione) '*EXT';



*inlr = *on;

